Integration 2

Exercise 1

```
> p[0] := sqrt(1/2);
  p[1] := sqrt(3/2) * x;
  p[2] := sqrt(5/8) * (3*x^2 - 1);
  p[3] := sqrt(7/8) * (5*x^3 - 3*x);
```

$$p_0 := \frac{\sqrt{2}}{2}$$

$$p_1 := \frac{\sqrt{6}\, x}{2}$$

$$p_2 := \frac{\sqrt{10}\,(3\,x^2 - 1)}{4}$$

$$p_3 := \frac{\sqrt{14}\,(5\,x^3 - 3\,x)}{4} \tag{1}$$

We find experimentally that $\int_{-1}^{1} p_i(x)\, p_j(x)\, \mathrm{d}x$ is 0 unless $i = j$, in which case the integral is 1.

Here are three cases:

```
> int(p[1] * p[2],x=-1..1);
```

$$0 \tag{2}$$

```
> int(p[2] * p[2],x=-1..1);
```

$$1 \tag{3}$$

```
> int(p[3] * p[2],x=-1..1);
```

$$0 \tag{4}$$

We can do all 16 cases together and print the results in a table, as follows:

```
> Matrix(
   [seq(
     [seq(int(p[i] * p[j],x=-1..1),
          j=0..3)],
    i=0..3)]
  );
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \tag{5}$$

We now need to find a function $p_4(x) = a\,x^4 + b\,x^2 + c$ that makes the pattern continue:

```
> p[4] := a*x^4 + b*x^2 + c ;
```

$$p_4 := a\,x^4 + b\,x^2 + c \tag{6}$$

The following integrals should be 0, 0, 0, 0 and 1, respectively:

```
> int(p[0] * p[4],x=-1..1);
  int(p[1] * p[4],x=-1..1);
  int(p[2] * p[4],x=-1..1);
  int(p[3] * p[4],x=-1..1);
  int(p[4] * p[4],x=-1..1);
```

$$\frac{\sqrt{2}\,a}{5} + \frac{\sqrt{2}\,b}{3} + \sqrt{2}\,c$$

$$0$$

$$\frac{4\sqrt{10}\,a}{35} + \frac{2\sqrt{10}\,b}{15}$$

$$0$$

$$\frac{2}{9}\,a^2 + \frac{4}{7}\,b\,a + \frac{4}{5}\,c\,a + \frac{2}{5}\,b^2 + \frac{4}{3}\,c\,b + 2\,c^2 \tag{7}$$

We can get Maple to find a, b and c based on this information:

```
> _EnvExplicit := true;
```

$$\_EnvExplicit := true \tag{8}$$

```
> sol := solve({
   int(p[0] * p[4],x=-1..1) = 0,
   int(p[1] * p[4],x=-1..1) = 0,
   int(p[2] * p[4],x=-1..1) = 0,
   int(p[3] * p[4],x=-1..1) = 0,
   int(p[4] * p[4],x=-1..1) = 1,
   a > 0
  },{a,b,c});
```

$$sol := \left\{a = \frac{105\sqrt{2}}{16},\, b = -\frac{45\sqrt{2}}{8},\, c = \frac{9\sqrt{2}}{16}\right\} \tag{9}$$

```
> p[4] := factor(subs(sol,p[4]));
```

$$p_4 := \frac{3\sqrt{2}\,\left(35\,x^4 - 30\,x^2 + 3\right)}{16} \tag{10}$$

```
> q := (n) -> sqrt(n + 1/2)*diff((x^2-1)^n,x$n)/(2^n * n!);
```

$$q := n \rightarrow \frac{\sqrt{n + \frac{1}{2}}\,\left(\frac{\partial^n}{\partial x^n}\,\left(x^2 - 1\right)^n\right)}{2^n\,n!} \tag{11}$$

```
> q(1); q(2); q(3); q(4);
```

$$\frac{\sqrt{6}\ x}{2}$$

$$\frac{\sqrt{10}\ (12\ x^2 - 4)}{16}$$

$$\frac{\sqrt{14}\ (48\ x^3 + 72\ (x^2 - 1)\ x)}{96}$$

$$\frac{\sqrt{2}\ \left(384\ x^4 + 1152\ (x^2 - 1)\ x^2 + 144\ (x^2 - 1)^2\right)}{256} \tag{12}$$

```
> seq(factor(q(n)/p[n]),n=1..4);
```

$$1, 1, 1, 1 \tag{13}$$

Exercise 2

The general picture is that for any positive integers n and m, there are constants a_0, ..., a_m such that

$$\int x^n \ln(x)^m \, dx = x^{n+1} \left(a_0 + a_1 \ln(x) + \ldots + a_m \log(x)^m\right)$$

Here are some examples:

```
> factor(int(x^3*ln(x)^5,x));
```

$$\frac{x^4\ (128 \ln(x)^5 - 160 \ln(x)^4 + 160 \ln(x)^3 - 120 \ln(x)^2 + 60 \ln(x) - 15)}{512} \tag{14}$$

```
> factor(int(x^4*ln(x),x));
```

$$\frac{x^5\ (5 \ln(x) - 1)}{25} \tag{15}$$

```
> factor(int(x^4*ln(x)^3,x));
```

$$\frac{x^5\ (125 \ln(x)^3 - 75 \ln(x)^2 + 30 \ln(x) - 6)}{625} \tag{16}$$

Exercise 3

These first few integrals involve functions that you may not have seen before. The erf( ) function is essentially the cumulative probability function for the normal distribution, so is important in statistics. The functions Ei( ), EllipticF( ) and hypergeom( ) are more obscure, but they have various applications in more advanced mathematics. You can learn more about them by entering ?Ei, ?EllipticF or ?hypergeom in Maple.

(a)

```
> int(exp(-x^2),x);
```

$$\frac{\sqrt{\pi}\ \mathrm{erf}(x)}{2} \tag{17}$$

(b)

```
> int(1/ln(x),x);
```

(18)

$$-\mathrm{Ei}_1(-\ln(x)) \tag{18}$$

(c)

```
> int(1/(sqrt(1-x^2)*sqrt(1-2*x^2)),x);
```

$$\mathrm{EllipticF}\left(x, \sqrt{2}\right) \tag{19}$$

(d)

```
> int((x^8+1)^(-1/2),x);
```

$$x \,\mathrm{hypergeom}\left(\left[\frac{1}{8}, \frac{1}{2}\right], \left[\frac{9}{8}\right], -x^8\right) \tag{20}$$

(e)

```
> int(sin(x)*ln(ln(x)),x);
```

$$\int \sin(x) \ln(\ln(x))\, \mathrm{d}x \tag{21}$$

(f)

```
> int(sin(sin(sin(x))),x);
```

$$\int \sin(\sin(\sin(x)))\, \mathrm{d}x \tag{22}$$

(g)

```
> int(1/sqrt(1+x+x^10),x);
```

$$\int \frac{1}{\sqrt{x^{10}+x+1}}\, \mathrm{d}x \tag{23}$$

We are looking for a function of x that can be defined in three characters, but cannot easily be integrated. At least one of the characters must be `x` (otherwise our function would be constant, so we could integrate it). That does not leave much space for anything else: we could have a letter or digit, or any of the characters `+`, `-`, `*`, `/` or `^`. Most functions built from these ingredients are easy to integrate:

```
> int(x+3,x);
```

$$\frac{1}{2} x^2 + 3\, x \tag{24}$$

```
> int(x*x,x);
```

$$\frac{x^3}{3} \tag{25}$$

```
> int(2^x,x);
```

$$\frac{2^x}{\ln(2)} \tag{26}$$

There is just one exception:

```
> int(x^x,x);
```

$$\int x^x\, \mathrm{d}x \tag{27}$$

There is actually an answer that is even shorter:

```
> int(x!,x);
```

(28)

$$\int x!\,dx \tag{28}$$

This is a little bit dubious because $x!$ is usually only defined when x is an integer, and that would make the integral meaningless. However, Maple does have a way to interpret $x!$ even when x is not an integer (eg $1.5! = 1.329340388$), which you can learn about by entering ?factorial and ?GAMMA.

Exercise 4

```
> y := sin(x) + sin(3*x)/3 + sin(5*x)/5 + sin(7*x)/7;
```

$$y := \sin(x) + \frac{\sin(3\,x)}{3} + \frac{\sin(5\,x)}{5} + \frac{\sin(7\,x)}{7} \tag{29}$$

```
> z := int(y,x);
```

$$z := -\frac{\cos(3\,x)}{9} - \frac{\cos(5\,x)}{25} - \frac{\cos(7\,x)}{49} - \cos(x) \tag{30}$$

The graph of y spends half the time wiggling near 0.8 and half the time wiggling near -0.8, jumping rapidly between these levels at $x = 0$, $x = \mathrm{Pi}$, $x = 2\,\mathrm{Pi}$ and all other multiples of Pi. The graph of z is a sawtooth, climbing from about -1.2 at $x = 0$ in a roughly straight line to about 1.2 at $x = \mathrm{Pi}$, then dropping in a roughly straight line to about -1.2 again at $x = 2\,\mathrm{Pi}$, and so on. This makes sense because $y = \frac{dz}{dx}$, which is the slope of the graph of z. When y is near 0.8, the graph of z slopes upwards at a roughly constant angle; when y is near -0.8, the graph of z slopes down at a nearly constant angle.

```
> plot([y,z],x=-2*Pi..2*Pi);
```




```
> unassign('y','z');
```

Exercise 5

```
> y := x*sin(20*x)*exp(-x);
```

$$y := x \sin(20\,x)\,e^{-x} \tag{31}$$

```
> z := 20*int(y,x);
```

$$z := 20\left(-\frac{20\,x}{401} - \frac{40}{160801}\right)e^{-x}\cos(20\,x) + 20\left(-\frac{x}{401} + \frac{399}{160801}\right)e^{-x}\sin(20\,x) \tag{32}$$

In this plot, y and z look very similar:

```
> plot([y,z],x=0..10);
```



When we look more closely, we see that size of the oscillations in the two graphs are nearly the same, as is the frequency of the oscillations, but the oscillations in y lag behind those in z.

```
> plot([y,z],x=0..3);
```



By focusing still further, we see that the extent of the lag is about a quarter of a complete cycle.

```
> plot([y,z],x=1.255..1.57);
```



All this makes sense if we look at the formulae:

```
> y;
```

$$x \sin(20\,x)\, e^{-x} \tag{33}$$

```
> z;
```

$$20\left(-\frac{20\,x}{401}-\frac{40}{160801}\right) e^{-x} \cos(20\,x)+20\left(-\frac{x}{401}+\frac{399}{160801}\right) e^{-x} \sin(20\,x) \tag{34}$$

If we expand out z, the first term is $-\left(\frac{400}{401}\right) x\, e^{-x} \cos(20\,x)$, which is nearly $-x\, e^{-x} \cos(20\,x)$. The remaining terms are much smaller. The function $-\cos(20\,x)$ is the same as $\sin(20\,x)$ shifted by a quarter cycle (because $\sin\left(\theta-\frac{\mathrm{Pi}}{2}\right)=-\cos(\theta)$). Thus, everything fits together.

In the first step mentioned above, you might want to enter **expand(z)** rather than doing it by hand. This does not work very well, as Maple insists on converting $\cos(20\,x)$ into a polynomial in $\cos(x)$, giving a very messy result. However, the following (rather obscure) syntax does the trick:

```
> evalf(frontend('expand',[z]));
```

$$-0.9975062344\, x \cos(20.\, x)\, e^{-1.\, x}-0.004975093438\, e^{-1.\, x} \cos(20.\, x) - 0.04987531172\, x \sin(20.\, x)\, e^{-1.\, x}+0.04962655705 \sin(20.\, x)\, e^{-1.\, x} \tag{35}$$

We now consider y^2+z^2:

```
> plot(y^2+z^2,x=0..6);
```




```
> simplify(y^2+z^2);
```

$$-\frac{1}{160801}\left(1201\left(\left(x^2-\frac{959200}{481601}x+\frac{63040400}{193122001}\right)\cos(20x)^2-\frac{16000\left(x+\frac{2}{401}\right)\sin(20x)\left(x-\frac{399}{401}\right)\cos(20x)}{1201}-\frac{161201x^2}{1201}+\frac{319200x}{481601}-\frac{63680400}{193122001}\right)e^{-2x}\right) \quad \textbf{(36)}$$

The first term is $\left(\frac{25921282001}{25856961601}\right)x^2 e^{-2x}$. As 25921282001 is close to 25856961601, this is approximately $x^2 e^{-2x}$. The other terms are much smaller, so we guess that y^2+z^2 should be close to $x^2 e^{-2x}$. We can check this graphically as follows:

```
> plot([y^2+z^2,x^2*exp(-2*x)],x=0..6);
```




```
> unassign('y','z');
```

Exercise 6

Here is a solution in a single step:

```
> solve(
    {seq(int(x^k*(a*x^2+b*x+c)*exp(-x),x=0..infinity)=k,k=1..3)},
    {a,b,c}
  );
```

$$\left\{a = -\frac{1}{4}, b = \frac{3}{2}, c = -\frac{1}{2}\right\} \quad \textbf{(37)}$$

Here it is again, more slowly:

```
> y := (a*x^2+b*x+c)*exp(-x);
```

$$y := \left(a\,x^2 + b\,x + c\right)\,\mathrm{e}^{-x} \quad \textbf{(38)}$$

```
> int1 := int(x*y,x=0..infinity);
```

$$int1 := 6\,a + 2\,b + c \quad \textbf{(39)}$$

```
> int2 := int(x^2*y,x=0..infinity);
```

$$int2 := 24\,a + 6\,b + 2\,c \quad \textbf{(40)}$$

```
> int3 := int(x^3*y,x=0..infinity);
```

$$int3 := 120\,a + 24\,b + 6\,c \quad \textbf{(41)}$$

```
> solve({int1 = 1, int2 = 2, int3 = 3},{a,b,c});
```
$$\left\{a = -\frac{1}{4}, b = \frac{3}{2}, c = -\frac{1}{2}\right\} \tag{42}$$

```
> unassign('y');
```

Exercise 7

```
> fsolve(x*(ln(ln(x)) + ln(x) - 1) = 541,{x=100});
```
$$\{x = 104.3555657\} \tag{43}$$

Exercise 8

```
> seq(a+b^n,n=2..10);
```
$$b^2 + a, b^3 + a, b^4 + a, b^5 + a, b^6 + a, b^7 + a, b^8 + a, b^9 + a, b^{10} + a \tag{44}$$

Exercise 9

```
> with(plots):
> display(
   implicitplot(x^4+y^4=1,x=-2..2,y=-2..2),
   plot([cos(t)/(2+cos(4*t)),sin(t)/(2+cos(4*t)),t=0..2*Pi])
  );
```



Exercise 10

```
> evalf[100](Pi^76/exp(87));
```

0.99947146427865260848967877132000132448482053488147209919429906118981421761076\
60153226662908744 1688961 **(45)**

Exercise 11

```
> simplify(sin(2*x) * tan(2*x) * diff(log(tan(x)),x,x));
```

$$-4 \tag{46}$$